
ΑΣΚΗΣΗ 7 

 
ΥΠΟΡΟΥΤΙΝΕΣ ΠΟΛΛΑΠΛΑΣΙΑΣΜΟΥ ΚΑΙ ΔΙΑΙΡΕΣΗΣ  

 
ΠΕΙΡΑΜΑΤΙΚΟ ΜΕΡΟΣ 

 
1.α) Το παρακάτω πρόγραμμα πολλαπλασιάζει το περιεχόμενο του καταχωρητή 

D με το περιεχόμενο του καταχωρητή E και αποθηκεύει το χαμηλό byte του α-
ποτελέσματος στον καταχωρητή L και το υψηλό στον H. Το πρόγραμμα αυτό 

υλοποιεί την πράξη του πολλαπλασιασμού με συνεχείς προσθέσεις δηλαδή προ-
σθέτει το περιεχόμενο του καταχωρητή E (πολλαπλασιαστέος) τόσες φορές όσες 

είναι το περιεχόμενο του καταχωρητή D (πολλαπλασιαστής). Στην αρχή το πρό-

γραμμα εξετάζει εάν το περιεχόμενο του καταχωρητή D είναι μηδέν, οπότε σε 
αυτή την περίπτωση βάζει στον καταχωρητή HL περιεχόμενο μηδέν. 

Αφού πρώτα αποθηκεύσετε στις θέσεις μνήμης 1230H και 1231H το περιεχόμε-
νο FF να εκτελέσετε το πρόγραμμα εντολή προς εντολή και με βάση τα αποτε-

λέσματα του εξομοιωτή να συμπληρώσετε τον παρακάτω πίνακα.  
 

Assembly Language Machine Language Bytes Cycles States 

START:  LD DE,(1230H) 0000:      

               CALL MULTI     

               LD (1232H),HL     

               HALT     

MULTI:   LD HL,0000H     

               LD A,D     

               OR A     

               RET Z     

               LD B,D     

               LD D,H     

LOOP:    ADD HL,DE     

               DJNZ  LOOP     

               RET     

 

1.β) Να αναφέρετε ποια είναι τα τελικά περιεχόμενα των καταχωρητών και της 

μνήμης όταν ολοκληρωθεί η εκτέλεση του προγράμματος. 
1.γ) Να υπολογισθεί ο συνολικός αριθμός των Bytes που καταλαμβάνει το πρό-

γραμμα στη μνήμη. 
1.δ) Να υπολογισθεί ο συνολικός αριθμός των States καθώς και ο συνολικόs 

χρόνος που απαιτείται για την εκτέλεση του προγράμματος εάν η συχνότητα 
του clock με το οποίο δουλεύει ο επεξεργαστής είναι 4 MHz.  

 
2.α) Το παρακάτω πρόγραμμα είναι μια υπορουτίνα γρήγορου πολλαπλασια-

σμού δύο δεδομένων που το καθένα είναι οκτώ bit. Το πρόγραμμα πολλαπλα-
σιάζει το περιεχόμενο του καταχωρητή D με το περιεχόμενο του καταχωρητή E 

και αποθηκεύει το χαμηλό byte του αποτελέσματος στον καταχωρητή L και το 

υψηλό στον H. Το πρόγραμμα αυτό για την υλοποίηση της πράξης του πολλα-

πλασιασμού ολισθαίνει το περιεχόμενο του καταχωρητή E (πολλαπλασιαστέος) 

συνολικά οκτώ φορές επειδή ο καταχωρητής D (πολλαπλασιαστής) έχει οκτώ bit 
και το προσθέτει στο περιεχόμενο του καταχωρητή HL τόσες φορές όσες είναι 

το πλήθος των bit του καταχωρητή D που έχουν τιμή ένα. Στην αρχή το πρό-
γραμμα εξετάζει εάν το περιεχόμενο του καταχωρητή D είναι μηδέν, οπότε σε 

αυτή την περίπτωση βάζει στον καταχωρητή HL περιεχόμενο μηδέν. 
Αφού πρώτα αποθηκεύσετε στις θέσεις μνήμης 1230H και 1231H το περιεχόμε-

νο FF να εκτελέσετε το πρόγραμμα εντολή προς εντολή και με βάση τα αποτε-
λέσματα του εξομοιωτή να συμπληρώσετε τον παρακάτω πίνακα.  

 
Assembly Language Machine Language Bytes Cycles States 

START:  LD DE,(1230H) 0000:      

               CALL MULTI     

               LD (1232H),HL     

               HALT     

MULTI:   LD HL,0000H                    

               LD A,D     

               OR A     

               RET Z     

               LD B,08H                    

               LD D,00H                    

LOOP1:  RRCA     

               JP NC,LOOP2     

               ADD HL,DE     

LOOP2:  SLA E     

               RL D     

               DJNZ LOOP1     

               RET     

 

2.β) Να αναφέρετε ποια είναι τα τελικά περιεχόμενα των καταχωρητών και της 
μνήμης όταν ολοκληρωθεί η εκτέλεση του προγράμματος. 

2.γ) Να υπολογισθεί ο συνολικός αριθμός των Bytes που καταλαμβάνει το πρό-
γραμμα στη μνήμη. 

2.δ) Να υπολογισθεί ο συνολικός αριθμός των States καθώς και ο συνολικόs 

χρόνος που απαιτείται για την εκτέλεση του προγράμματος εάν η συχνότητα 
του clock με το οποίο δουλεύει ο επεξεργαστής είναι 4 MHz.  

 
3.α) Το παρακάτω πρόγραμμα είναι μια υπορουτίνα γρήγορου πολλαπλασια-

σμού δύο δεδομένων που το καθένα είναι δέκα έξι bit. Το πρόγραμμα πολλα-
πλασιάζει τα περιεχόμενα των καταχωρητών BC με τα περιεχόμενα των κατα-

χωρητών DE και αποθηκεύει τα δύο χαμηλά byte του αποτελέσματος στους κα-
ταχωρητές HL και τα δύο υψηλά στους DE.  

Το πρόγραμμα αυτό για την υλοποίηση της πράξης του πολλαπλασιασμού ολι-
σθαίνει τα περιεχόμενα των καταχωρητών DE HL συνολικά δέκα έξι φορές επει-

δή το ζευγάρι των καταχωρητών DE που αντιπροσωπεύει τον πολλαπλασιαστή 
είναι συνολικά δέκα έξι bit. Σε κάθε μία ολίσθηση το πρόγραμμα ελέγχει το α-

ντίστοιχο bit του πολλαπλασιαστή. Εάν αυτό είναι ένα τότε προσθέτει τα περιε-
χόμενα των καταχωρητών BC που αντιπροσωπεύουν τον πολλαπλασιαστέο με 

τα περιεχόμενα των καταχωρητών HL, ενώ εάν αυτό είναι μηδέν τότε εκτελεί 

την επόμενη ολίσθηση των καταχωρητών DE HL.  

Αφού πρώτα αποθηκεύσετε στις θέσεις μνήμης 1230H, 1231H, 1232H, 1233H 
το περιεχόμενο FF να εκτελέσετε το πρόγραμμα εντολή προς εντολή και με βά-

ση τα αποτελέσματα του εξομοιωτή να συμπληρώσετε τον παρακάτω πίνακα.  
 

Assembly Language Machine Language Bytes Cycles States 

START:  LD BC,(1230H) 0000:      

               LD DE,(1232H)     

               CALL MULTI     

               LD (1234H),HL     

               LD (1236H),DE     

               HALT     

MULTI:   LD HL,0000H                                  

               LD A,10H     

LOOP1:  ADD HL,HL     

               RL E         

               RL D     

               JP NC,LOOP2     

               ADD HL,BC     

               JP NC,LOOP2     

               INC DE     

LOOP2:  DEC A     

               JP NZ,LOOP1     

               RET     

 
3.β) Να αναφέρετε ποια είναι τα τελικά περιεχόμενα των καταχωρητών και της 

μνήμης όταν ολοκληρωθεί η εκτέλεση του προγράμματος. 
3.γ) Να υπολογισθεί ο συνολικός αριθμός των Bytes που καταλαμβάνει το πρό-

γραμμα στη μνήμη. 

3.δ) Να υπολογισθεί ο συνολικός αριθμός των States καθώς και ο συνολικόs 
χρόνος που απαιτείται για την εκτέλεση του προγράμματος εάν η συχνότητα 

του clock με το οποίο δουλεύει ο επεξεργαστής είναι 4 MHz.  
 

4.α) Το παρακάτω πρόγραμμα διαιρεί το περιεχόμενο των καταχωρητών HL με 
το περιεχόμενο του καταχωρητή E και αποθηκεύει το πηλίκο στους BC και το 

υπόλοιπο στους HL. Το πρόγραμμα αυτό υλοποιεί την πράξη της διαίρεσης με 
συνεχείς αφαιρέσεις δηλαδή αφαιρεί από το περιεχόμενο των καταχωρητών HL 

(διαιρετέος) το περιεχόμενο του καταχωρητή E (διαιρέτης) και κάθε φορά που η 
διαφορά είναι μεγαλύτερη του μηδενός δηλαδή ο μειωμένος διαιρετέος είναι με-

γαλύτερος του διαιρέτη αυξάνει το περιεχόμενο του BC (πηλίκο) κατά ένα. Ε-
πειδή υπάρχει περίπτωση το περιεχόμενο του διαιρετέου να είναι ίσο με το πε-

ριεχόμενο του διαιρέτη για να έχουμε σε αυτή την περίπτωση το πηλίκο ίσο με 
ένα πρέπει να βάλουμε για αρχική τιμή του καταχωρητή BC το FFFF δηλ. το -1 

και όχι το μηδέν. Βάζοντας το μηδέν το πρόγραμμα μας βγάζει ότι το πηλίκο της 

διαίρεσης είναι ίσο με δύο και όχι ίσο με ένα που είναι το σωστό αποτέλεσμα. 
Αφού πρώτα αποθηκεύσετε στις θέσεις μνήμης 1230H, 1231H και 1232H το πε-

ριεχόμενο FF να εκτελέσετε το πρόγραμμα εντολή προς εντολή και με βάση τα 
αποτελέσματα του εξομοιωτή να συμπληρώσετε τον παρακάτω πίνακα.  

Assembly Language Machine Language Bytes Cycles States 

START:  LD HL,(1230H) 0000:      

               LD A,(1232H)     

               AND A     

               JP Z,END     

               LD E,A     

               CALL DIV8     

               LD (1233H),BC     

               LD (1235H),HL     

END:      HALT     

DIV8:      LD BC,0FFFFH     

               LD D,00H     

LOOP:    SBC HL,DE     

               INC BC     

               JP NC,LOOP     

               ADD HL,DE        

               RET     

 

4.β) Να αναφέρετε ποια είναι τα τελικά περιεχόμενα των καταχωρητών και της 
μνήμης όταν ολοκληρωθεί η εκτέλεση του προγράμματος. 

4.γ) Να υπολογισθεί ο συνολικός αριθμός των Bytes που καταλαμβάνει το πρό-
γραμμα στη μνήμη. 

4.δ) Να υπολογισθεί ο συνολικός αριθμός των States καθώς και ο συνολικόs 

χρόνος που απαιτείται για την εκτέλεση του προγράμματος εάν η συχνότητα 
του clock με το οποίο δουλεύει ο επεξεργαστής είναι 4 MHz.  

 
5.α) Το παρακάτω πρόγραμμα είναι μία υπορουτίνα γρήγορης διαίρεσης ενός 

αριθμού δέκα έξι bit με ένα αριθμό οκτώ bit. Συγκεκριμένα διαιρεί το περιεχό-
μενο των καταχωρητών HL που είναι ο διαιρετέος με το περιεχόμενο του κατα-

χωρητή D που είναι ο διαιρέτης και αποθηκεύει το πηλίκο στον καταχωρητή L 
και το υπόλοιπο στον καταχωρητή A. Το πρόγραμμα αυτό υλοποιεί την πράξη 

της διαίρεσης χρησιμοποιώντας τους καταχωρητές A,H,L σαν ένα ενιαίο κατα-
χωρητή που έχει συνολικά εύρος είκοσι τέσσερα bit. Το πρόγραμμα ολισθαίνει 

τα δέκα έξι bit του καταχωρητή HL στον καταχωρητή A και ελέγχει τη δυνατό-
τητα διαιρετότητας συγκρίνοντας το περιεχόμενο του καταχωρητή A με το πε-

ριεχόμενο του καταχωρητή D. Εάν το περιεχόμενο του καταχωρητή A είναι μι-
κρότερο από το περιεχόμενο του καταχωρητή D τότε κάνει ολίσθηση του επό-

μενου bit του καταχωρητή HL. Εάν το περιεχόμενο του καταχωρητή A είναι με-

γαλύτερο ή ίσο από το περιεχόμενο του καταχωρητή D τότε αφαιρεί από τον A 
τον D και αυξάνει κατά ένα το πηλίκο, δηλαδή το περιεχόμενο του καταχωρητή 

L.  
Η διαδικασία αυτή επαναλαμβάνεται συνολικά δέκα έξι φορές όσα δηλαδή είναι 

τα bit του καταχωρητή HL που είναι ο διαιρετέος. 
Αφού πρώτα αποθηκεύσετε στις θέσεις μνήμης 1230H, 1231H και 1232H το πε-

ριεχόμενο FF να εκτελέσετε το πρόγραμμα εντολή προς εντολή και με βάση τα 
αποτελέσματα του εξομοιωτή να συμπληρώσετε τον παρακάτω πίνακα.  

Assembly Language Machine Language Bytes Cycles States 

START:  LD HL,(1230H) 0000:      

               LD A,(1232H)     

               AND A     

               JP Z,END     

               LD D,A     

               CALL DIV8     

               LD (1233H),HL     

               LD (1235H),A     

END:      HALT     

DIV8:      XOR A     

               LD B,10H     

LOOP1:  ADD HL,HL     

               RLA     

               CP D     

               JP C,LOOP2     

               SUB D     

               INC L     

LOOP2:  DJNZ LOOP1        

               RET     

 
5.β) Να αναφέρετε ποια είναι τα τελικά περιεχόμενα των καταχωρητών και της 

μνήμης όταν ολοκληρωθεί η εκτέλεση του προγράμματος. 
5.γ) Να υπολογισθεί ο συνολικός αριθμός των Bytes που καταλαμβάνει το πρό-

γραμμα στη μνήμη. 
5.δ) Να υπολογισθεί ο συνολικός αριθμός των States καθώς και ο συνολικόs 

χρόνος που απαιτείται για την εκτέλεση του προγράμματος εάν η συχνότητα 

του clock με το οποίο δουλεύει ο επεξεργαστής είναι 4 MHz.  
 


